
Тема занятия: Работа в электронных таблицах
ИНСТРУКЦИЯ ПО ВЫПОЛНЕНИЮ ЗАДАНИЙ:
1. Ознакомьтесь с дополнительным материалом, приведенным ниже:
- https://resh.edu.ru/subject/lesson/5817/main/82481/ 
- https://www.youtube.com/watch?v=JNPXnu96NCw

Обратная связь: ei-kapustina@yandex.ru    

Теоретическая часть
Функции
Функция - это программа, которая выполняет различные вычисления. Каждая функция имеет собственный синтаксис, за которым необходимо следить, чтобы программа работала корректно. Ввод функции начинаются со знака =. После этого знака располагается имя функции, а затем один или несколько аргументов, заключенных в скобки. 
Например: =СУММ(D5:D11)  - суммирует значения ячеек в диапазоне D5:D11
Функции в Excel
	Категория
	назначение

	Финансовые
	Вычисляет процентные ставки, ежемесячные отчисления, амортизационные отчисления

	Дата и время
	Отображает текущий час, день недели, время или дату

	Математические и тригонометрические
	Вычисляют абсолютные величины, косинусы и логарифмы

	Статические
	Вычисляют средние значения, наибольшие и наименьшие числа в диапазоне, коэффициенты распределения; тестируют на независимость выборок

	Ссылки и массивы
	Вычисляют и возвращают значения из диапазона; создают гиперссылки для сетевых или Web-документов

	Работа с базами данных
	Вычисляют значения в таблицах баз данных Excel

	Логические
	Вычисляют выражения и возвращают значения ИСТИНА или ЛОЖЬ, которые используются при выполнении другого действия или форматирования.

	Информационные
	Возвращают из Excel вWindows информацию о текущем статусе ячейки, объекта или среды

	Инженерные
	Входят в состав Excel , но должны устанавливаться отдельно из Analysis Toolpack

	Текстовые
	Преобразуют регистр текста, образуют символы с правого или левого конца текстовой строки, объединяют текстовые строки


· помощью функций обработки текста можно выполнять различные преобразования над строковыми данными.В Microsoft Excel имеется 24 функции, относящихся к данной группе. Рассмотрим использование некоторых из них при решении различных задач, часто возникающих в практической деятельности:
СЦЕПИТЬ (текст1; текст2; …) – объединяет несколько текстовых строк в одну.
СЖПРОБЕЛЫ (текст) – удаляет из текста лишние пробелы (кроме одиночных пробелов между словами).
ЛЕВСИМВ (текст;кол_зн) – возвращает указанное количество знаков с начала строки текста.
НАЙТИ (строка;текст; поз) – возвращает номер позиции первого вхождения строки в текст, начиная с указанной позиции. Нумерация ведется относительно левого символа текста.
ДЛСТР(текст) – возвращает количество знаков в текстовой строке.
ПРАВСИМВ(текст;кол_зн) – возвращает указанное количество знаков с конца строки текста.
ПСТР ( текст; поз; кол_зн) – возвращает заданное количество знаков из строки текста, начиная с указанной позиции.
ЗАМЕНИТЬ (текст; поз; кол_зн; строка) – заменяет в тексте начиная с указанной позиции заданное количество символов на другую строку.

Ввод функций
1. Перед вводом функции убедитесь, что ячейка для размещения является активной.
2. Вызовите Мастер функций: лента Формулы –Вставить функцию fx .
3. Выберите в списке нужную категорию, а ниже– функцию, в нижней части окна прочтите описание функции. Щелкните на кнопке ОК.
4. В окне Аргументы функции заполните необходимые поля, щелкните на кнопке ОК.
Коды ошибок появляются в ячейках при обнаружении конфликтов.
	Код и название ошибки
	Причина

	#####
	Данные слишком велики, чтобы поместить в ячейке. Проверьте формулу или увеличьте ширину столбца

	#дел/0
	Число или содержимое ячейки, на которое вы делите, равно нулю или ячейка пуста. Если такое происходит с только что заполненными ячейками, то, скорее всего, нужна абсолютная ссылка на ячейку в исходной формуле.

	#н/д
	Вы пропустили обязательный параметр функции, или ячейка, содержащая данные, пуста или содержит данные не того типа, который требуется

	#имя?
	Вы неверно указали имя диапазона или функции, или вы использовали ссылку на несуществующее имя, или вы использовали в формуле текст, не заключив его в кавычки, или вы пропустили двоеточие при задании диапазона

	#пусто!
	Вы указали пересечение, которого на самом деле не существует

	#число!
	Вы указали текстовый или пустой параметр там, где требуется число, или вы ввели формулу, которая при вычислении выдает число, слишком большое или слишком малое для Excel

	#ссылка!
	Вы удалили ячейки, необходимые формуле, так что Excel не может найти необходимую ячейку (вы могли удалить ячейки и путем вставки на их место других ячеек)

	#знач!
	Вы ввели текст, там где требуется числовое значение или логический оператор ИСТИННО или ЛОЖНО.


2.Выполните  задания:

ЗАДАНИЕ №1
Подготовьте таблицу для расчетов заработной платы сотрудников предприятия за октябрь. Выполните расчеты. Если число детей сотрудника три и более, то подоходный налог не взимается. Выполните сортировку сотрудников по алфавиту. Рассчитайте размер средней заработной платы сотрудников.
Образец таблицы.
[image: ]
ХОД РАБОТЫ:
ПОСТРОЕНИЕ ТАБЛИЦЫ
1. [image: ]Выделите диапазон ячеек А1:Н1. 
2. Объедините их –лента Главная – группа Выравнивание – Объединить и поместить в центре
3. Внесите заголовок таблицы и нажмите Enter
4. Выделите диапазон ячеек А2:Н16.
5.  Поставьте границы - лента Главная – группа Шрифт – Границы – Все границы[image: ]
6. Выделите диапазон А2:А3, объедините, внесите данные № п/п
7. Повторите действия для диапазонов В2:В3, С2:С3, D2:D3, заполните их.
8. Выделите диапазон Е2:G2, объедините, внесите данные Налоги
9. Заполните ячейки E3,F3,G3, используя данные из образца
10. Выделите диапазон Н2:Н3, объедините, внесите данные Сумма к выдаче
11. Заполните ячейки E4,F4,G4, используя данные из образца
12. Внесите в ячейку А5 цифру 1, в ячейку А6 цифру 2
13. Выделите две эти ячейки, наведите указатель мыши на нижний правый угол ячейки А6.
14. Указатель примет вид черного крестика. Используйте автозаполнение. (Протяните вниз или дважды щелкните по нему)
15. Заполните таблицу данными из образца по столбцам ФИО, Число детей и Оклад.
16. [image: ][image: ]Выделите диапазон D5:Н15. Выставите денежный формат – лента Главная – группа Число – Денежный (используйте значок или выберите из списка )

17. Выделите диапазон ячеек А2:Н3.
18. [image: ]Установите для этих ячеек перенос текста по словам: лента Главная – группа Выравнивание – Перенос текста

19. Выделите диапазон А2:Н16. Измените размеры столбцов: 
1) лента Главная – группа Ячейки – Формат – Автоподбор ширины столбца
20. Выделите столбцы Е, F, G и установите для ни ширину 15: 
1) [image: http://www.pixempire.com/images/preview/great-arrow-down-icon.jpg]Щелкните правой кнопкой мыши по имени столбца Е.  Указатель примет вид толстой стрелочки, направленной вниз 
2) удерживая кнопку мыши нажатой, протяните до столбца F 
ПРИМЕЧАНИЕ: Или щелчок правой кнопкой мыши по столбцу Е. Нажмите и удержите клавишу SHIFT  щелкните по столбцу  F
3) Лента Главная – группа Ячейки – Формат – Ширина столбца
4) Наберите с клавиатуры число 15 и нажмите ОК
ПРИМЕЧАНИЕ: Если часть заголовков не отображается, то увеличьте высоту строки 3. Наведите указатель мыши на границу между номерами строк 3 и 4, удерживая правую кнопку мыши нажатой,  потяните вниз. Как только все записи проявятся, отпустите правую кнопку мыши. 
21. Измените ширину столбца С согласно образцу. Выделите столбец. Наведите указатель мыши на границу между столбцами С и D.  Указатель примет вид двойной стрелочки. [image: http://www.compbegin.ru/data/image/excel_7.png] Удерживая левую кнопку мыши потяните влево, пока столбец не примет вид образца. 
ВЫПОЛНЕНИЕ РАСЧЕТОВ
1. Рассчитайте размер Профсоюзного взноса по формуле  Профсоюз = Оклад * % Премии, в ячейке E5 наберите формулу = D5*$Е$4 (ячейка E4 используется в виде абсолютной адресации) и скопируйте автозаполнением
ПРИМЕЧАНИЕ: Ввод формулы начинается со знака =. Чтобы не вводить адрес ячейки вручную, щелкните мышью по соответствующей ячейке, ее адрес пропишется в формуле
2. Аналогично рассчитайте Пенсионный взнос.
3. Для расчета подоходного налога используем стандартную функцию =ЕСЛИ( ) рис. 1
1) В ячейку G5 введем формулу, используя мастер функций: лента Формулы – группа Библиотека функция – Логические – Если
2) В диалоговом окне Аргументы функции в строке Логическое выражение внесите условие проверки количества детей сотрудника C5>=3 (если число детей больше или равно трем)
3) В строке Значение_если_истина ставим 0 (т.к. если детей больше или равно трем, то налог не взимается)
4) В строке Значение_если_ложь ставим формулу для расчета подоходного налога D5*$G$4 и нажмите ОК

5) Скопируйте формулу автозаполнением
6) Измените количество детей у сотрудников. Посмотрите, как сработает формула.
7) Отмените изменения
4. Расчет Сумма к выдачи выполните по формуле: Сумма к выдачи = Оклад- Профсоюз- Пенсионный взнос - Подоходный налог (=D5-E5-F5-G5)
5. Скопируйте ее автозаполнением для остальных сотрудников
6. Расчет итоговых значений по столбцам Налоги и Сумма к выдачи выполните с помощью автосуммы:
1) Выделите диапазон Е5:Е15
2) Лента Главная – Редактирование – Автосумма[image: ]
7. Распространите ее автозаполнением на остальные столбцы
8. Расчет Средней заработной платы сотрудника выполните с помощью формулы СРЗНАЧ()
9. Выделите диапазон F18:G18, объедините ячейки.
10. Внесите запись Средняя заработная плата
11. В ячейку Н18 вставьте формулу для расчета:
1) Лента Формулы – группа Библиотека функция – Другие функции – Статистические – СРЗНАЧ
2) В диалоговом окне Аргументы функции в строке Число 1 внесите диапазон H5:H16, выделив его из таблицы и нажмите ОК. рис 2
[image: ]
СОРТИРОВКА
1. [image: ]Для сортировки сотрудников по алфавиту выделите диапазон В5:Н15 (выделение только столбца с фамилиями при сортировке перемешает данные в таблице)
2. Лента Главная – Редактирование – Сортировка и фильтр – Сортировка от А до Я

ЗАДАНИЕ №2
Подготовьте таблицу для расчетов заработной платы сотрудников предприятия ноябрь. Исправьте размер профсоюзного взноса с 1% на 2%, Добавьте доплату сотрудникам в размере 10% от оклада.
Образец таблицы
[image: ]
ХОД РАБОТЫ
[image: ]СОЗДАНИЕ ТАБЛИЦЫ
Таблица практически идентична таблице из задания 2. Чтобы не строить ее заново, скопируем лист целиком и внесем в него изменения в соответствии с образцом.
1. Переименуйте ярлычок Листа 1, присвоив ему имя «Зарплата октябрь» Для этого дважды щелкните мышью по ярлычку и наберите новое имя, нажмите Enter.
2. Скопируйте содержимое листа «Зарплата октябрь» на новый лист. 
1) Нажмите правой кнопкой мыши по ярлыку листа «Зарплата октябрь» 
2) Выберите Переместить/Скопировать 
3) Выберите перед листом: Лист 2
4) Поставьте галочку в окошке Создавать копию. рис. 3
ПРИМЕЧАНИЕ. Перемещать и копировать листы можно, перетаскивая их корешки (для копирования удерживайте нажатой клавишу CTRL). 
3. Выделите столбец Е, 
4. Добавьте столбец слева от столбца Е.
1) Лента Главная – группа Ячейки – Вставить – Вставить столбцы на лист
5. Внесите заголовок столбца  Доплата и размер доплаты 5%
ВЫПОЛНЕНИЕ РАСЧЕТОВ
1. Вычислите размер доплаты формуле  Доплата = Оклад * % Доплаты
2. Скопируйте ее автозаполнением для остальных сотрудников
3. Исправьте формулу Суммы к выдаче. К исходным результатам нужно добавить доплату:
1) Выделите ячейку I5
2) Перейдите в Строку формул
3) Прибавьте к исходным результатам значение ячейки Е5, нажмите Enter.
4. Скопируйте ее автозаполнением для остальных сотрудников
5. Выполните расчет итоговой суммы Доплаты в ячейке Е16 любым из известных способов (Автозаполнение, Автосумма или расчет по формуле) 

ЗАДАНИЕ №3. Запустите электронную таблицу (программу Microsoft Excel).
1. Лист 1 переименуйте в Расчёты и создайте таблицу по образцу, внеся в таблицу не менее 7 работников бригады.
[image: ]









Рекомендации по заполнению таблицы:
− Колонки «Фамилия и инициалы», «Должность» и «КТУ» заполните произвольно. − «КТУ» - это коэффициент трудового участия.
−  Начисленная	зарплата,	например,	в	ячейке	E4,	считается	по	формуле:
= D4 * $E$2 / СУММ (D$4: D$…)
− Знак $ перед буквой и цифрой указывает на то, что они являются абсолютными и не изменяются при копировании формулы.
− Вместо абсолютной ссылки $E$2 можно использовать имя, присвоенное этой ячейке, например, «всего». Тогда формула в ячейке E4 будет выглядеть так:


−  Скопируйте эту формулу в ячейки ниже.
[image: ]
3. Переименуйте второй лист рабочей книги в «СЛАУ».
−  Для переименования листа щёлкните по его ярлыку правой клавишей мыши и выберите команду «Переименовать». Введите новое имя листа. Можно использовать команду Главная → Формат → Переименовать Лист.
4. Наберите с помощью вставки формул заданное уравнение (согласно варианту) по образцу:
[image: ]


Варианты к заданию:


[image: ]





























ЗАДАНИЕ №4. У классного руководителя есть список учащихся класса в следующем виде
	
	A
	B
	C

	1
	Агеев
	Андрей
	Иванович

	2
	…
	…
	…

	3
	Яковлев
	Яков
	Яковлевич



Но ему нужно представить список несколько в другом виде, объединив фамилию, имя и отчество в одной ячейке:
	
	A
	B
	C

	1
	Агеев Андрей Иванович
	
	

	2
	…
	
	…

	3
	Яковлев Яков Яковлевич
	
	


Решение.
Эта задача легко решается с помощью одной функции СЦЕПИТЬ. Выделив ячейку D1, с помощью Мастера функций в категории Текстовые выбираем эту функцию и заполняем
нужные поля следующим образом (адреса ячеек вводятся автоматически, если щелкнуть мышью по соответствующей клетке в таблице; кавычки набирать не нужно, они также появятся автоматически после ввода пробела и перехода к следующему полю!)

Осталось только растянуть формулу на нужное количество строк. В столбце D список получен . Правда, если удалить столбцы А:С, то список «разрушится», вместо нужных данных, в столбце появятся ошибочные значения #ССЫЛКА! – мы удалили ячейки, на которые были ссылки в формулах. Чтобы такого не произошло, нужно сначала получить новый список в ячейках в виде значений (а не формул). Для этого нужно выделить столбец D и скопировать его (в буфер). Затем выделить нужный столбец, диапазон или верхнюю ячейку диапазона и в меню Правка выбрать пункт Специальная вставка. В открывшемся окне выбрать радиокнопку Значения и нажать ОК.
В выбранном диапазоне формулы заменятся на значения и полученный список может «существовать» совершенно независимо от исходного.
ЗАДАНИЕ №5.Необходимо получить из списка
	
	A
	
	B
	C

	1
	Фамилия
	Имя
	Отчество

	2
	Агеев
	Андрей
	Иванович

	3
	…
	…
	…

	4
	Яковлев
	Яков
	Яковлевич

	список фамилий с инициалами:
	

	
	
	
	
	

	
	A
	
	B
	C

	1
	Фамилия И.О.
	
	
	

	2
	Агеев А.И.
	
	
	

	3
	…
	
	
	…

	4
	Яковлев Я.Я.
	
	
	



Решение.
Выполним поставленную задачу последовательно . Сначала в ячейку D1 внесем формулу с функцией, которая выделяет первый символ из имени. Для этого с помощью Мастера функций в категории Текстовые выберем функцию ЛЕВСИМВ (рекомендуется в практической работе и дальше все функции вставлять в формулы с использованием Мастера функций).
[image: ]










После сохранения формулы в ячейке D1 скопируем ее в ячейку E1 для выделения первого символа отчества. А затем с помощью функции СЦЕПИТЬ объединим нужные ячейки, пробел и точки.
· помощью копирования и специальной вставки сохраним полученный список в виде значений в нужном диапазоне.
Задача 3 (обратная задаче 1).
Из имеющегося списка
	
	A
	B
	C

	1
	Агеев Андрей Иванович
	
	

	2
	…
	
	…

	3
	Яковлев Яков Яковлевич
	
	


нужно получить список, в котором фамилии имена и отчества расположены в отдельных ячейках.
	
	A
	B
	C

	1
	Фамилия
	Имя
	Отчество

	2
	Агеев
	Андрей
	Иванович

	3
	…
	…
	…

	4
	Яковлев
	Яков
	Яковлевич


Решение.
При решении данной задачи будем считать, что между фамилией, именем и отчеством стоят ровно по одному пробелу. Если это не так, сначала приведем исходный список к такой структуре. Чтобы не усложнять задачу, будем выполнять действия последовательно,
записывая в отдельные ячейки промежуточные результаты расчетов с использованием различных функций:
1) В ячейку В1 – формулу, позволяющую удалить лишние пробелы. = СЖПРОБЕЛЫ(А1)
2) в ячейку С1 – формулу, позволяющую найти позицию первого пробела
3) =НАЙТИ(" ";B1) – третий параметр не указан, т.к. ищем первое вхождение пробела от начала строки.
4) В ячейку D1 – формулу, позволяющую найти позицию второго пробела (между именем и отчеством)
5) =НАЙТИ(" ";B1;C1+1) – здесь третий параметр указывает, что поиск ведется со следующего символа от найденного в С1 пробела.
6) В ячейку E1 – формулу для определения общего количества символов в фамилии, имени и отчестве, включая два пробела.
7) =ДЛСТР(B1)
8) В ячейку F1 – формулу для выделения фамилии. Воспользуемся уже знакомой функцией. =ЛЕВСИМВ(B1;C1-1)
9) Ячейку пока пропустим.
10) В ячейку H1 – формулу для выделения отчества. Отчество расположено в конце строки, поэтому воспользуемся функцией ПРАВСИМВ, а количество выделяемых символов легко найдем, зная номер позиции пробела перед отчеством и общую длину строки. =ПРАВСИМВ(B1;E1-D1)
11) В ячейку G1 – формулу для выделения имени. Имя расположено внутри строки между пробелами, позиции которых мы уже вычислили. Для его выделения воспользуемся функцией
12) =ПСТР(B1;C1+1;D1-C1-1)
13) С помощью копирования и специальной вставки сохраним полученный список в виде значений.
ЗАДАНИЕ №6.Из имеющегося списка
	
	A
	
	B
	
	C

	1
	Агеев Андрей Иванович
	
	
	
	
	

	2
	…
	
	
	
	
	…

	3
	Яковлев Яков Яковлевич
	
	
	
	
	

	нужно получить список фамилий с инициалами
	
	

	
	
	
	
	
	

	
	A
	B
	
	
	C

	1
	Агеев А.И.
	
	
	
	
	

	2
	…
	
	
	
	…

	3
	Яковлев Я.Я.
	
	
	
	
	



Решите данную задачу самостоятельно, используя изученные функции.
ЗАДАНИЕ №7. У секретаря учебной части имеется список учащихся выпускного класса в следующем виде:
	
	A
	B
	C

	1
	Агеев
	Андрей
	Иванович

	2
	…
	…
	…

	3
	Яковлев
	Яков
	Яковлевич


Для печати аттестатов нужно получить аналогичный список учащихся в дательном падеже. Конечно, можно вручную изменять окончания в каждой ячейке, но если выпускников много, то это длительный процесс. Помогите автоматизировать эту работу, переведя в дательный падеж по возможности большую часть имеющихся фамилий, имен и отчеств, используя возможности Excel.
Решение.
Конечно, очень трудно определить все возможные типы фамилий, имен и отчеств, встречающиеся у российских граждан, поэтому определим как меняются окончания в наиболее типичных фамилиях и именах в именительном и дательном падежах. Так как в русском языке склонение существительных и прилагательных различаются в мужском и женском роде, то и основные типы женских и мужских фамилий и имен представим в отдельных таблицах.


	Женские фамилии в
	Женские фамилии в
	
	Женские фамилии в
	Женские фамилии
	

	именительном падеже
	дательном падеже
	
	именительном
	в дательном
	

	
	
	
	падеже
	падеже
	

	На -ова, -ева, -ина:
	Петровой, Агеевой,
	
	На -ия :
	Марии
	

	Петрова, Агеева, Катина
	Катиной
	
	Мария
	
	

	На -ая:
	
	
	На -ья:
	Дарье
	

	Римская, Оболонная
	Римской, Оболонной
	
	Дарья
	
	

	На -о:
	Хомченко
	
	На согласные:
	Мариам, Дениз
	

	Хомченко
	
	
	Мариам, Дениз
	
	

	На -а: Водяха
	Водяха
	
	На -а: Татьяна
	Татьяне
	

	На -ь: Козарь
	Козарь
	
	На -ь: Нинель
	Нинели
	

	На -ик: Бублик
	Бублик
	
	
	
	

	На -ых: Пьяных
	Пьяных
	
	
	
	


В женском отчестве на -овна или -евна в дательном падеже буква «а» заменяется на «е».

	Мужские фамилии в
	Мужские фамилии в
	
	Мужские имена
	Мужские имена в
	

	именительном падеже
	дательном падеже
	
	в именительном
	дательном падеже
	

	
	
	
	падеже
	
	

	На -ов, -ев, -ин:
	Петрову, Агееву, Катину
	
	На -ей, -ий:
	Сергею, Дмитрию
	

	Петров, Агеев, Катин
	
	
	Сергей, Дмитрий
	
	

	На -о: Хомченко
	Хомченко
	
	На -я: Илья
	Илье
	

	На -ий, -ый:
	
	
	На согласные:
	Глебу, Ивану
	

	Римский, Оболонный
	Римскому, Оболонному
	
	Глеб, Иван
	
	

	На -а: Водяха
	Водяхе
	
	На -а: Никита
	Никите
	

	На -ь: Козарь
	Козарю
	
	На -ь: Игорь
	Игорю
	

	На -ик: Бублик
	Бублику
	
	
	
	

	На -ых: Пьяных
	Пьяных
	
	
	
	


К мужскому отчеству на -ович или -евич в дательном падеже добавляется буква «у».
Чтобы не слишком усложнять формулы для перевода фамилий, имен и отчеств из именительного в дательный падеж, следует разделить списки по полу, и сделать преобразования для каждого списка по-отдельности. Будет проще, если в соседнем с отчеством столбце добавить соответственно символы «ж» и «м» и отсортировать список по этому столбцу. Затем скопировать каждую часть списка на отдельные листы ЭТ.
Начнем преобразование с женских фамилий. Из таблицы видно, что если фамилия оканчивается на «ая», то эти две буквы нужно заменить на «ой», если на «ва» или «на», то на «ой» нужно заменить только последнюю букву «а». В остальных случаях фамилия не изменяется.
Если все выделения букв, проверки и замены учесть в одной формуле, она будет слишком громоздкой, поэтому в ячейках D1: G1 выполним промежуточные действия.
	
	Формула
	Пояснение

	D1
	=ДЛСТР(A1)
	количество символов в фамилии (длину строки)

	E1
	=ПРАВСИМВ(A1;2)
	две последние буквы фамилии

	F1
	=ЗАМЕНИТЬ(A1;D1-1;2;"ой")
	замена двух последних букв в фамилии

	G1
	=ЗАМЕНИТЬ(A1;D1;1;"ой")
	замена одной последней буквы в фамилии


Тогда в ячейке H1 из этих частей фамилия в дательном падеже будет сформирована с помощью следующей формулы:
=ЕСЛИ(E1="ая";F1;ЕСЛИ(ИЛИ(E1="ва";E1="на");G1;A1))

Преобразование женских имен выполним другим способом – выделяя часть имени и сцепляя с нужными окончаниями. В ячейках I1: L1 выполним промежуточные действия
	Ячейка
	Формула
	Пояснение

	I1
	=ДЛСТР(B1)
	количество символов в имени (длина строки)

	J1
	=ПРАВСИМВ(B1;1)
	последняя буква имени

	K1
	=ЛЕВСИМВ(B1;I1-1)
	часть имени без последней буквы

	L1
	=ПРАВСИМВ(B1;2)
	две последние буквы имени


В ячейке М1 имя может быть сформировано следующей формулой:
=ЕСЛИ(ИЛИ(J1="а";L1="ья");СЦЕПИТЬ(K1;"е");ЕСЛИ(L1="ия";СЦЕПИТЬ(K1;"и");B1))

· ячейке N1 сформируем дательный падеж отчества: =ЗАМЕНИТЬ(C1;ДЛСТР(C1);1;"е")
Теперь выделим диапазон всех заполненных ячеек D1:N1 и скопируем (растяним) их на весь список. Сохраним столбцы с полученными фамилиями, именами и отчествами как значения в нужном диапазоне на этом же или другом листе электронной таблицы.
Преобразование мужских фамилий несколько сложнее, так как нужно учесть большее количество вариантов. Поэтому выполним промежуточные действия.
	Ячейка
	Формула
	Пояснение

	D1
	=ДЛСТР(A1)
	количество символов в фамилии

	E1
	=ПРАВСИМВ(A1)
	последняя буква фамилии

	F1
	=ПРАВСИМВ(A1;2)
	две последние буквы фамилии

	G1
	=ЛЕВСИМВ(A1;D1-1)
	часть фамилии без последней буквы


· следующих ячейках в каждой формуле будем не только проводить вычисления с помощью различных функций для всех типов фамилий, но в каждой ячейке будем формировать фамилии определенных типов, при этом последовательно объединяя их. Это будет особенно наглядно, когда формулы будут скопированы на весь список. Например, в ячейке Н1 запишем формулу для формирования дательного падежа фамилий с окончанием на «а» и «ь», в ячейке I1 – для фамилий с окончанием на «ий» и «ый», в ячейке J1- учтем фамилии с окончанием на «ых» и «ко» (они не изменяются), а в ячейке К1 добавим остальные типы мужских фамилий, у которых в дательном падеже появляется окончание «у».
	H1
	=ЕСЛИ(E1="а";СЦЕПИТЬ(G1;"е");ЕСЛИ(E1="ь";СЦЕПИТЬ(G1;"ю");""))

	I1
	=ЕСЛИ(ИЛИ(F1="ий";F1="ый");ЗАМЕНИТЬ(A1;D1-1;2;"ому");H1)

	J1
	=ЕСЛИ(ИЛИ(F1="ых";F1="ко");A1;I1)

	K1
	=ЕСЛИ(J1<>"";J1;СЦЕПИТЬ(A1;"у"))


В следующих ячейках сформируйте самостоятельно имена и отчества.

КОНТРОЛЬНЫЕ ВОПРОСЫ

1. Что такое функция? Как использовать мастер функций?
2. Как указать ячейки в формуле?
3. Что такое Автозаполнение?
4. Каким способом в Excel можно соединить 2 или более строки?
5. В чем заключается отличие функции ЗАМЕНИТЬ() от функции ПОДСТАВИТЬ()?
6. В чем заключается отличие функции НАЙТИ() от функции ПОИСК()?
7. Какая функция позволяет избавиться от лишних пробельных символов?
8. Какие операторы ссылки используются в Microsoft Excel?
9. В чем состоит особенность использования функции Автосумма?
10. Опишите способы копирования формул в Microsoft Excel.
11. Объясните различия между относительными и абсолютными ссылками.
12. Охарактеризуйте работу с вложенными функциями в Microsoft Excel.

Ответы на вопросы на свободный лист в документе Excel.

Желаю удачи и здоровья!
Ваш преподаватель: Капустина Елена Ивановна

